

## 01

```
PREFIX i: <http://nosql.opendata.cz/school/terms#>
SELECT ?n ?f ?l
FROM <http://nosql.opendata.cz/school/>
WHERE {
  ?s rdf:type i:Student ;
  i:number ?n ;
  i:name ?b .
  ?b i:first ?f ; i:last ?l .
}

... WHERE {
  ?s rdf:type i:Student .
  ?s i:number ?n .
  ?s i:name ?b .
  ?b i:first ?f .
  ?b i:last ?l .
}

... WHERE {
  ?s rdf:type i:Student ;
  i:number ?n ;
  i:name [ i:first ?f ; i:last ?l ] .
}

... WHERE {
  ?s rdf:type i:Student ;
  i:number ?n ;
  i:name _:abc .
  _:abc i:first ?f ; i:last ?l .
}

... SELECT $n $f $l
... WHERE {
  $s rdf:type i:Student ;
  i:number $n ;
  i:name [ i:first $f ; i:last $l ] .
}
```

## 02

```
PREFIX i: <http://nosql.opendata.cz/school/terms#>
SELECT ?c ?t
FROM <http://nosql.opendata.cz/school/>
WHERE {
  ?p rdf:type i:Course ;
  i:code ?c ;
  i:title ?t .
  FILTER ( regex(?c, "^NPRG") )
}
ORDER BY ?t

... FILTER regex(?c, "^NPRG") ...

... ORDER BY ASC(?t)

... [ rdf:type i:Course ; i:code ?c ; i:title ?t . ] ...

... [] rdf:type i:Course ; i:code ?c ; i:title ?t . ...

... _:abc rdf:type i:Course ; i:code ?c ; i:title ?t . ...
```

### 03

```
PREFIX i: <http://nosql.opendata.cz/school/terms#>
SELECT ?n ?e
FROM <http://nosql.opendata.cz/school/>
WHERE {
  ?s rdf:type i:Student ;
  i:number ?n .
  OPTIONAL { ?s i:email ?e . }
}
```

## 04

```
PREFIX i: <http://nosql.opendata.cz/school/terms#>
SELECT ?n ?e ?w
FROM <http://nosql.opendata.cz/school/>
WHERE {
  ?s rdf:type i:Student ;
  i:number ?n .
  OPTIONAL { ?s i:email ?e }
  OPTIONAL { ?s i:web ?w }
}
```

Incorrect:

```
... WHERE {
  ?s rdf:type i:Student ;
  i:number ?n
  OPTIONAL { ?s i:email ?e ; i:web ?w }
}
```

Incorrect:

```
... WHERE {
  ?s rdf:type i:Student ;
  i:number ?n .
  OPTIONAL {
    ?s i:email ?e
    OPTIONAL { ?s i:web ?w }
  }
}
```

```

PREFIX i: <http://nosql.opendata.cz/school/terms#>
PREFIX s: <http://nosql.opendata.cz/school/>
SELECT DISTINCT ?p ?c
FROM <http://nosql.opendata.cz/school/>
WHERE {
  ?p rdf:type i:Course ; i:code ?c .
  ?e i:course ?p ; i:day ?d ; i:semester s:semester171 .
  FILTER ( (?d = "MON") || (?d = "FRI") )
}

... WHERE {
  ?p rdf:type i:Course ; i:code ?c .
  [ i:course ?p ; i:day ?d ; i:semester s:semester171 ]
  FILTER ( (?d = "MON") || (?d = "FRI") )
}

... WHERE {
  ?p rdf:type i:Course ; i:code ?c .
  ?e i:course ?p ; i:day ?d ; i:semester ?s .
  FILTER ( (?d = "MON") || (?d = "FRI") )
  FILTER ( ?s = s:semester171 )
}

... WHERE {
  {
    ?p rdf:type i:Course ; i:code ?c .
    ?e i:course ?p ; i:day ?d ; i:semester s:semester171 .
    FILTER ( ?d = "MON" )
  } UNION {
    ?p rdf:type i:Course ; i:code ?c .
    ?e i:course ?p ; i:day ?d ; i:semester s:semester171 .
    FILTER ( ?d = "FRI" )
  }
}

... WHERE {
  {
    ?p rdf:type i:Course ; i:code ?c .
    ?e i:course ?p ; i:day "MON" ; i:semester s:semester171 .
  } UNION {
    ?p rdf:type i:Course ; i:code ?c .
    ?e i:course ?p ; i:day "FRI" ; i:semester s:semester171 .
  }
}

```

```

PREFIX i: <http://nosql.opendata.cz/school/terms#>
PREFIX s: <http://nosql.opendata.cz/school/>
SELECT ?p ?c
FROM <http://nosql.opendata.cz/school/>
WHERE {
  ?p rdf:type i:Course ; i:code ?c .
  FILTER NOT EXISTS {
    ?e i:course ?p ; i:day ?d ; i:semester s:semester171 .
    FILTER ( (?d = "MON") || (?d = "FRI") )
  }
}

```

```

PREFIX i: <http://nosql.opendata.cz/school/terms#>
PREFIX s: <http://nosql.opendata.cz/school/>
SELECT ?p ?c
FROM <http://nosql.opendata.cz/school/>
WHERE {
  ?p rdf:type i:Course ; i:code ?c .
  MINUS {
    ?e i:course ?p ; i:day ?d ; i:semester s:semester171 .
    FILTER ( (?d = "MON") || (?d = "FRI") )
  }
}

```

Incorrect:

```

PREFIX i: <http://nosql.opendata.cz/school/terms#>
PREFIX s: <http://nosql.opendata.cz/school/>
SELECT DISTINCT ?p ?c
FROM <http://nosql.opendata.cz/school/>
WHERE {
  ?p rdf:type i:Course ; i:code ?c .
  ?e i:course ?p ; i:day ?d ; i:semester s:semester171 .
  FILTER ( (?d != "MON") && (?d != "FRI") )
}

```

## 07

```
PREFIX i: <http://nosql.opendata.cz/school/terms#>
PREFIX s: <http://nosql.opendata.cz/school/>
SELECT ?p ?c
FROM <http://nosql.opendata.cz/school/>
WHERE {
  ?p rdf:type i:Course ; i:code ?c .
  OPTIONAL {
    ?e i:course ?p ; i:day ?d ; i:semester s:semester171 .
    FILTER ( (?d = "MON") || (?d = "FRI") )
  }
  FILTER ( ! bound(?e) )
}

... WHERE {
  ?p rdf:type i:Course ; i:code ?c .
  OPTIONAL {
    ?e1 i:course ?p ; i:day "MON" ; i:semester s:semester171
  }
  OPTIONAL {
    ?e2 i:course ?p ; i:day "FRI" ; i:semester s:semester171
  }
  FILTER ( ! bound(?e1) && ! bound(?e2) )
}
```

## 08

```
PREFIX i: <http://nosql.opendata.cz/school/terms#>
PREFIX s: <http://nosql.opendata.cz/school/>
SELECT ?f ?l (AVG(?r) AS ?a)
FROM <http://nosql.opendata.cz/school/>
WHERE {
  ?s rdf:type i:Student ;
  i:name [ i:first ?f ; i:last ?l ] ;
  i:enroll [ i:course ?c ; i:result ?r ; i:semester s:semester171 ] .
}
GROUP BY ?s ?f ?l
HAVING (COUNT(?c) <= 10)
```